
by Steven W. Vagts
Editor, “Z-100 LifeLine”

ZBASIC

FROGMATH.BAS

FROGMATH.BAS is a program originally based upon
"NUMBERS.BAS" from Gerry Kabelman and described
in an article in Issue #20, September 1981, of
"REMark" magazine. Gerry's picture of his little
girl playing the game made the cover of that
issue.

In 1981, my oldest little girl was 8 and needed
some math drill programs. And I thought of the
NUMBERS.BAS program.

Originally written in MBASIC, I rewrote it for
BASIC-E to run on my H-88. And to add a little
interest, I added a frog, a spider, and a fly -
and my little girl would sit for hours before
the computer, though I did let her out of the
chains (er ... chair) for nourishment and sleep.
The other two daughters used it also.

For a “Z-100 LifeLine” contest in 1996, I took
the program out of retirement and rewrote it for
the Z-100's ZBASIC, adding color and some motion
to the spider and the fly.

The program was originally published in issue
#46, from August 1996 and I have now added it to
the Website.

If you have a youngster just learning math, give
this program a try.

FROGMATH.BAS

10   REM -------------------------------------------------------------------- 
20   REM     FROGMATH   -   COMPUTER AIDED INSTRUCTION   -  7/96 
30   REM                by  Steven W. Vagts 
40   REM                211 Sean Way, Hendersonville, NC  28792
50   REM     ZBASIC program for the H/Z-100            Updated April 2022 
60   REM -------------------------------------------------------------------- 
100  DIM L$(10) 
200  REM Set ESCAPE Codes. 
210  E$=CHR$(27): F$=E$+"F": G$=E$+"G": REM E$=ESCape, F$=Graphics On, G$=Off 
220  P$=E$+"p": Q$=E$+"q": REM P$=Reverse Video On, Q$=Off 
230  FP$=F$+P$: GQ$=G$+Q$ 
400  REM  Define graphic parts of numerals. 
410  N1$=FP$+"ppp"+GQ$: N2$=" "+FP$+"p"+Q$+" "+G$: N3$=F$+"pp"+P$+" "+GQ$ 
420  N4$=FP$+"pp "+GQ$: N5$=FP$+" "+Q$+"pp"+G$: N6$=FP$+" "+Q$+"p"+P$+" "+GQ$ 
430  N7$=FP$+" p "+GQ$: N8$=FP$+" "+Q$+" "+P$+" "+GQ$: N9$=FP$+"p p"+GQ$ 
440  N10$=FP$+" "+GQ$: N11$=N10$+"  ": N12$=FP$+"ppp"+GQ$ 
450  M1$=FP$+"r"+Q$+"r_"+P$+"_"+GQ$: M2$=F$+"_"+P$+"_r"+Q$+"r"+G$ 
460  M3$=FP$+"r p"+GQ$: M4$=FP$+"r"+Q$+"p"+P$+"_"+GQ$ 
470  M5$=F$+"_"+P$+"p"+Q$+"r"+G$: M6$=FP$+"r "+GQ$ 
480  M7$=FP$+" pp"+GQ$: M8$=" "+FP$+"r"+Q$+"r"+G$ 
490  M9$=FP$+"r"+Q$+"r "+G$ 
500  COLOR 7,0,0: CLS 
600  REM  Opening screen 
610  COLOR 2,0: LOCATE 10,20: PRINT "  "+FP$+"r@p"+GQ$+" " 
620  LOCATE 11,20: PRINT " "+FP$+"r   "+GQ$+" " 
630  LOCATE 12,19: PRINT " "+FP$+"r   y_"+GQ$+" " 
640  LOCATE 13,18: PRINT " "+FP$+"pa y"+Q$+"r "+P$+" p"+GQ$+" " 
650  COLOR 7,0 
700  LOCATE 7,31: PRINT N5$+" "+N6$+" "+M4$; 
705  PRINT " "+M4$ 
710  LOCATE 8,31: PRINT N5$+" "+FP$+" p"+Q$+"r "+N8$+" "+FP$+" "+Q$+"l"; 
715  PRINT P$+"p"+GQ$ 
720  LOCATE 9,31: PRINT N11$+" "+N10$+F$+"_"+P$+"_"+Q$+" "; 
725  PRINT M5$+F$+" _"+P$+"p "+GQ$ 
730  LOCATE 11,30: PRINT FP$+" _r "+Q$+"  "+P$+"r_"+Q$+"  p"; 
735  PRINT P$+" "+Q$+"p "+N8$ 
740  LOCATE 12,30: PRINT N10$+F$+"_r"+N10$+" "+FP$+"r"+Q$+"r_"; 
745  PRINT P$+"_"+Q$+"  "+N11$+N6$ 
750  LOCATE 13,30: PRINT N11$+FP$+" "+Q$+" "+P$+" "+Q$+"pp"; 
755  PRINT P$+" "+Q$+"  "+N11$+N8$ 
760  LOCATE 17,45: PRINT "by  Steven W. Vagts" 
770  LOCATE 18,49: PRINT "211 Sean Way" 
780  LOCATE 19,49: PRINT "Hendersonville, NC  28792" 
790  LOCATE 24,1: INPUT "Press {RETURN} to continue", A$ 
1000 REM  FROGMATH Program 
1010 CLS: PRINT TAB(15);"CAI - COMPUTER AIDED INSTRUCTION - MATH": PRINT 
1020 PRINT: INPUT "Hello, my name is Fritz.  What is your name"; NME$ 
1030 PRINT: PRINT "Hi, "+NME$+".  ": INPUT "Can you tell time"; A$ 
1040 IF LEFT$(A$,1)="Y" OR LEFT$(A$,1)="y" GOTO 1070 
1050 PRINT: INPUT "Ok. Instead, press any number or letter for me. ", A$ 
1060 GOTO 1100 
1070 PRINT: INPUT "Great!  What time is it"; A$ 
1100 PRINT: PRINT "Thank you, "+NME$: RANDOMIZE ASC(A$): PRINT 
1110 PRINT "The object of this game is to `ZAP' the spider and the fly" 
1120 PRINT "in the fewest number of moves.  For each correct answer, you" 
1130 PRINT "get to either move the frog or fire his tongue." 
1140 PRINT "The game ends upon `ZAPping' both.": PRINT 
1150 PRINT "You may select the degree of difficulty from 1 to 4." 
1160 PRINT TAB(10);"1 = EASY                  4 = HARD" 
1170 PRINT: INPUT "How hard do you want these problems"; HARD$ 
1180 IF HARD$="" OR LEN(HARD$)>1 GOTO 1150 
1190 IF ASC(HARD$)<49 OR ASC(HARD$)>52 GOTO 1150 
1200 PRINT: PRINT "There are 5 types of operations." 
1210 PRINT TAB(10);"1 = ADDITION" 
1220 PRINT TAB(10);"2 = SUBTRACTION" 
1230 PRINT TAB(10);"3 = MULTIPLICATION" 
1240 PRINT TAB(10);"4 = DIVISION" 
1250 PRINT TAB(10);"5 = Random mixed" 
1260 PRINT: INPUT "Which number do you select"; TYPE$ 

2



1270 IF TYPE$="" OR LEN(TYPE$)>1 GOTO 1260 
1280 IF ASC(TYPE$)<49 OR ASC(TYPE$)>53 GOTO 1260 
1290 MOVE=0: NUM=0: MSG=1: FIN1=0: FIN2=0: W=0: DIR=6 
1300 FROG=20: PFLY=50: PSPD=20: SPD=1 
1500 REM  Start problems 
1510 PSX=10: PSY=34: NUM=NUM+1: W1=0: T=VAL(TYPE$): CLS 
1520 IF T>4 THEN T=INT(4*RND)+1 
1530 IF T>4 GOTO 1520 
1540 IF HARD$="1" THEN J=10: J1=10 
1550 IF HARD$="2" THEN J=20: J1=20 
1560 IF HARD$="3" THEN J=100: J1=100 
1570 IF HARD$="4" THEN J=1000: J1=1000 
1580 IF T=3 THEN J1=10 
1590 IF T=4 THEN J1=10: J=5*VAL(HARD$) 
1600 X=INT(J*RND): Y=INT(J1*RND) 
1610 REM  Ensure no division by zero. 
1620 IF T=4 AND X=0 OR Y=0 GOTO 1600 
1630 REM  Ensure that for difficulty 1 or 2 that Y is less than X. 
1640 IF T=2 AND Y>X AND VAL(HARD$)<3 GOTO 1600 
1650 ON T GOTO 1660, 1670, 1680, 1690 
1660 Z=X+Y: GOTO 1710 
1670 Z=X-Y: GOTO 1710 
1680 Z=X*Y: GOTO 1710 
1690 Z1=X*Y: X=Z1: Z=X/Y 
1700 REM GOSUB to print problem - calculate length of answer. 
1710 CLS: IF FIN2=0 THEN GOSUB 6410: REM  Print spider 
1720 IF FIN1=0 THEN GOSUB 6110: REM  Print stopped fly 
1750 P=0: COLOR 3,0: V$=STR$(X): FOR I=LEN(V$) TO 2 STEP -1 
1760 V1=VAL(MID$(V$,I,1))+1: P=P+1: REM  I=1 is sign posit, V1 must be >0 
1770 IF P=1 THEN PNX=6: PNY=55 
1780 IF P=2 THEN PNX=6: PNY=50 
1790 IF P=3 THEN PNX=6: PNY=45 
1800 IF P=4 THEN PNX=6: PNY=40 
1810 ON V1 GOSUB 5601, 5611, 5621, 5631, 5641, 5651, 5661, 5671, 5681, 5691 
1820 NEXT I: P=0: V$=STR$(Y): FOR I=LEN(V$) TO 2 STEP -1 
1830 V1=VAL(MID$(V$,I,1))+1: P=P+1: REM  I=1 is sign posit, V1 must be >0 
1840 IF P=1 THEN PNX=10: PNY=55 
1850 IF P=2 THEN PNX=10: PNY=50 
1860 IF P=3 THEN PNX=10: PNY=45 
1870 IF P=4 THEN PNX=10: PNY=40 
1880 ON V1 GOSUB 5601, 5611, 5621, 5631, 5641, 5651, 5661, 5671, 5681, 5691 
1890 NEXT I: ON T GOSUB 5210, 5310, 5410, 5510 
1900 LOCATE 14,34: PRINT F$+STRING$(24,112)+G$ 
2000 REM  Print last position of frog. 
2010 IF DIR=6 THEN GOSUB 5810 
2020 IF DIR=4 THEN GOSUB 5910 
2030 V$=STR$(Z) 
2100 REM  Get answer. 
2110 COLOR 3,0: LOCATE 14,1,1: INPUT "ANSWER"; ANS$ 
2120 IF ANS$="" GOTO 2110 
2130 F1=1: FOR I=1 TO LEN(ANS$): REM  Test ANS$ for legal chars 
2140 IF MID$(ANS$,I,1)="-" AND I=1 THEN F1=2: GOTO 2160 
2150 IF ASC(MID$(ANS$,I,1))<48 OR ASC(MID$(ANS$,I,1))>57 GOTO 2110 
2160 NEXT I: ANS=VAL(ANS$): COLOR 3,0: P=0 
2170 FOR I=LEN(ANS$) TO F1 STEP -1: V1=VAL(MID$(ANS$,I,1))+1: P=P+1 
2180 IF P=1 THEN PNX=16: PNY=55 
2190 IF P=2 THEN PNX=16: PNY=50 
2200 IF P=3 THEN PNX=16: PNY=45 
2210 IF P=4 THEN PNX=16: PNY=40 
2220 ON V1 GOSUB 5601, 5611, 5621, 5631, 5641, 5651, 5661, 5671, 5681, 5691 
2230 NEXT I: IF ANS<0 THEN PSX=16: PSY=34: GOSUB 5310 
2240 GOSUB 5010 
2250 IF ANS<>Z THEN GOSUB 7010: GOTO 2110 
2500 REM  Print "CORRECT" message. 
2510 IF MSG=5 THEN MSG=1 
2520 LOCATE 16,1,1: ON MSG GOTO 2530, 2540, 2550, 2560, 2570 
2530 PRINT "You're right!": GOTO 2580 
2540 PRINT "Absolutely correct!": GOTO 2580 
2550 PRINT "Correct!": GOTO 2580 
2560 PRINT "Right, "+NME$+"!": GOTO 2580 
2570 PRINT "That was easy, wasn't it?" 
2580 IF W1=1 THEN GOSUB 5010: GOTO 2700 

2600 MOVE=MOVE+1: COLOR 3,0: LOCATE 25,4,0 
2610 PRINT "`4' = move left     `6' = move right     `8' fires tongue" 
2620 LOCATE 18,4,1: INPUT "What is your move <4, 6, or 8>"; ANS$ 
2630 IF ANS$="4" THEN FROG=FROG-1: DIR=4: GOSUB 5910: GOTO 2670 
2640 IF ANS$="6" THEN FROG=FROG+1: DIR=6: GOSUB 5810: GOTO 2670 
2650 IF ANS$="8" THEN GOSUB 7410: GOTO 2670 
2660 GOTO 2620 
2670 COLOR 3,0: LOCATE 25,4,0: PRINT SPC(57) 
2680 IF FIN1=1 AND FIN2=1 GOTO 2800 
2700 GOSUB 7320: LOCATE 16,1: ON MSG GOTO 2710, 2720, 2730, 2740, 2750 
2710 PRINT "Try this one.": GOTO 2760 
2720 PRINT "How about this one?": GOTO 2760 
2730 PRINT "Can you do this one?": GOTO 2760 
2740 PRINT "Now a tough one.": GOTO 2760 
2750 PRINT "Here's another." 
2760 GOSUB 5010: MSG=MSG+1: GOTO 1510 
2800 REM  Print score. 
2810 G=NUM-W: H=INT(G*100/NUM) 
2820 CLS: LOCATE 1,1,1: COLOR 7,0: PRINT "You scored ";H;"% CORRECT!" 
2830 IF PSPD>2 GOTO 2860 
2840 PRINT "It took ";MOVE;" moves to ZAP the fly and the spider escaped!" 
2850 PRINT "I hope to help you do better next time.": PRINT: GOTO 3000 
2860 PRINT "It took ";MOVE;" moves to ZAP the spider and the fly." 
2900 REM  Set parameters for congratulatory messages. 
2910 IF H>90 GOTO 2950 
2920 IF H>80 GOTO 2960 
2930 IF H>70 GOTO 2970 
2940 GOTO 3000 
2950 PRINT "EXCELLENT GAME, ";NME$: PRINT: GOTO 3000 
2960 PRINT "Very GOOD GAME, ";NME$: PRINT: GOTO 3000 
2970 PRINT "Not bad, ";NME$: PRINT 
3000 INPUT "Wish to try again"; ANS$ 
3010 IF LEFT$(ANS$,1)="Y" OR LEFT$(ANS$,1)="y" GOTO 1150 
3020 PRINT "OK, ";NME$;", another time.  Have a nice day." 
3030 STOP 
5000 REM   * * * SUBROUTINES * * * 
5010 FOR I=1 TO 1000: NEXT I: RETURN 
5200 REM  Print add sign: 
5210 LOCATE PSX,PSY: PRINT N2$ 
5220 LOCATE PSX+1,PSY: PRINT N9$ 
5230 LOCATE PSX+2,PSY+1: PRINT N10$: RETURN 
5300 REM  Print subtract sign: 
5310 LOCATE PSX+1,PSY: PRINT N1$: RETURN 
5400 REM  Print multiply sign: 
5410 LOCATE PSX+1,PSY-1: PRINT M2$ 
5420 LOCATE PSX+2,PSY-1: PRINT M1$: RETURN 
5500 REM  Print division sign: 
5510 LOCATE PSX,PSY: PRINT N2$ 
5520 LOCATE PSX+1,PSY: PRINT N1$ 
5530 LOCATE PSX+2,PSY: PRINT N2$: RETURN 
5600 REM  Print number 0: 
5601 LOCATE PNX,PNY: PRINT M4$ 
5602 LOCATE PNX+1,PNY: PRINT N8$ 
5603 LOCATE PNX+2,PNY: PRINT M5$: RETURN 
5610 REM  Print number 1: 
5611 LOCATE PNX,PNY: PRINT M6$ 
5612 LOCATE PNX+1,PNY+1: PRINT N10$ 
5613 LOCATE PNX+2,PNY: PRINT N9$: RETURN 
5620 REM  Print number 2: 
5621 LOCATE PNX,PNY: PRINT N6$ 
5622 LOCATE PNX+1,PNY: PRINT M8$ 
5623 LOCATE PNX+2,PNY: PRINT M3$: RETURN 
5630 REM  Print number 3: 
5631 LOCATE PNX,PNY: PRINT N3$ 
5632 LOCATE PNX+1,PNY: PRINT N3$ 
5633 LOCATE PNX+2,PNY: PRINT N4$: RETURN 
5640 REM  Print number 4: 
5641 LOCATE PNX,PNY: PRINT FP$+"r"+Q$+"r"+N10$ 
5642 LOCATE PNX+1,PNY: PRINT N7$ 
5643 LOCATE PNX+2,PNY+2: PRINT N10$: RETURN 
5650 REM  Print number 5: 
5651 LOCATE PNX,PNY: PRINT N5$ 

5652 LOCATE PNX+1,PNY: PRINT N3$ 
5653 LOCATE PNX+2,PNY: PRINT N4$: RETURN 
5660 REM  Print number 6: 
5661 LOCATE PNX,PNY: PRINT M9$ 
5662 LOCATE PNX+1,PNY: PRINT M7$ 
5663 LOCATE PNX+2,PNY: PRINT N7$: RETURN 
5670 REM  Print number 7: 
5671 LOCATE PNX,PNY: PRINT N6$ 
5672 LOCATE PNX+1,PNY: PRINT M8$ 
5673 LOCATE PNX+2,PNY: PRINT M9$: RETURN 
5680 REM  Print number 8: 
5681 LOCATE PNX,PNY: PRINT N6$ 
5682 LOCATE PNX+1,PNY: PRINT N6$ 
5683 LOCATE PNX+2,PNY: PRINT N7$: RETURN 
5690 REM  Print number 9: 
5691 LOCATE PNX,PNY: PRINT N6$ 
5692 LOCATE PNX+1,PNY: PRINT N3$ 
5693 LOCATE PNX+2,PNY+2: PRINT N10$: RETURN 
5800 REM  Print right-facing Frog 
5810 COLOR 2,0: LOCATE 20,FROG: PRINT "  "+FP$+"r@p"+GQ$+" " 
5820 LOCATE 21,FROG: PRINT " "+FP$+"r   "+GQ$+" " 
5830 LOCATE 22,FROG-1: PRINT " "+FP$+"r   y_"+GQ$+" " 
5840 LOCATE 23,FROG-2: PRINT " "+FP$+"pa y"+Q$+"r "+P$+" p"+GQ$+" " 
5850 COLOR 7,0: RETURN 
5860 REM  Print right frog's tongue. 
5870 COLOR 2,0: FOR I=0 TO 150: PSET (((FROG+4)*8)+I,174-I): NEXT 
5880 FOR T=1 TO 100: NEXT T: REM Time delay to see tongue, then delete 
5890 COLOR 0,0: FOR I=150 TO 0 STEP -1: PSET (((FROG+4)*8)+I,174-I): NEXT 
5895 RETURN 
5900 REM  Print left-facing Frog 
5910 COLOR 2,0: LOCATE 20,FROG: PRINT " "+FP$+"p@_"+GQ$+"  " 
5920 LOCATE 21,FROG: PRINT " "+FP$+"   _"+GQ$+" " 
5930 LOCATE 22,FROG-1: PRINT " "+FP$+"rx   _"+GQ$+" " 
5940 LOCATE 23,FROG-2: PRINT " "+FP$+"p "+Q$+" _"+P$+"x ap"+GQ$+" " 
5950 COLOR 7,0: RETURN 
5960 REM  Print left frog's tongue. 
5970 COLOR 2,0: FOR I=0 TO 150: PSET ((FROG*8)-I,174-I): NEXT 
5980 FOR T=1 TO 100: NEXT T:  REM Time delay to see tongue, then delete 
5990 COLOR 0,0: FOR I=150 TO 0 STEP -1: PSET ((FROG*8)-I,174-I): NEXT 
5995 RETURN 
6100 REM  Print Stopped Fly: 
6110 COLOR 6,0: LOCATE 2,PFLY 
6120 PRINT " "+F$+"y^x"+G$+" " 
6130 LOCATE 3,PFLY: PRINT "  "+N10$+"  " 
6140 LOCATE 4,PFLY-4: PRINT "     "+F$+"r _"+G$+"     ": RETURN 
6200 IF PFLY<10 THEN RETURN: REM  Print Left Flying Fly: 
6210 COLOR 6,0: LOCATE 2,PFLY: PRINT F$+" ^"+P$+"pr"+GQ$+" " 
6220 LOCATE 3,PFLY: PRINT F$+" x yy "+G$ 
6230 LOCATE 4,PFLY: PRINT "   Bzzzz ": RETURN 
6300 IF PFLY>65 THEN RETURN: REM  Print Right Flying Fly: 
6310 COLOR 6,0: LOCATE 2,PFLY: PRINT " "+FP$+"_p"+Q$+"^ "+G$ 
6320 LOCATE 3,PFLY: PRINT F$+" xx y "+G$ 
6330 LOCATE 4,PFLY: PRINT " Bzzzz   ": RETURN 
6400 REM  Print Spider: 
6410 COLOR 4,0: ON SPD GOTO 6420, 6450, 6500 
6420 LOCATE 2,PSPD: PRINT F$+"yyxx"+G$ 
6430 LOCATE 3,PSPD+1: PRINT FP$+":q"+GQ$ 
6440 LOCATE 4,PSPD: PRINT F$+"xxyy"+G$: SPD=SPD+1: RETURN 
6450 LOCATE 2,PSPD: PRINT F$+"yyyx"+G$ 
6460 LOCATE 3,PSPD+1: PRINT FP$+":q"+GQ$ 
6470 LOCATE 4,PSPD: PRINT F$+"xxxy"+G$: SPD=SPD+1: RETURN 
6500 PSPD=PSPD-1: SPD=1: REM  Move spider 
6510 LOCATE 2,PSPD: PRINT F$+"yxxx "+G$ 
6520 LOCATE 3,PSPD+1: PRINT FP$+":q"+GQ$+" " 
6530 LOCATE 4,PSPD: PRINT F$+"xyyy "+G$ 
6540 IF PSPD=2 THEN FIN2=1 ELSE RETURN 
6550 LOCATE 16,1: PRINT "The spider has escaped!" 
6560 LOCATE 2,1: PRINT SPC(6) 
6570 LOCATE 3,1: PRINT SPC(6) 
6580 LOCATE 4,1: PRINT SPC(6) 
6590 GOSUB 5010: GOTO 7330 
7000 REM  Print "INCORRECT" messages. 

7010 IF MSG=5 THEN MSG=1 
7020 IF W1=1 GOTO 7100 
7030 W1=1: W=W+1: COLOR 3,0: LOCATE 15,1 
7040 ON MSG GOTO 7050, 7060, 7070, 7080, 7090 
7050 PRINT "Sorry. That's incorrect. Try again.": GOTO 7200 
7060 PRINT "No, ";NME$;". Let's have another try.": GOTO 7200 
7070 PRINT "That's wrong. Take another look.": GOTO 7200 
7080 PRINT "Not so, ";NME$;". Here it is again.": GOTO 7200 
7090 PRINT "No. Think about it for awhile. I'm in no hurry.": GOTO 7200 
7100 GOSUB 7320: LOCATE 16,1,0: PRINT "The correct answer is" 
7110 COLOR 3,0: P=0: V$=STR$(Z) 
7120 IF Z<0 THEN PSX=6: PSY=29: GOSUB 5310 
7130 FOR I=LEN(V$) TO 2 STEP -1: V1=ASC(MID$(V$,I,1))-47: P=P+1 
7140 IF P=1 THEN PNX=16: PNY=55 
7150 IF P=2 THEN PNX=16: PNY=50 
7160 IF P=3 THEN PNX=16: PNY=45 
7170 IF P=4 THEN PNX=16: PNY=40 
7180 ON V1 GOSUB 5601, 5611, 5621, 5631, 5641, 5651, 5661, 5671, 5681, 5691 
7190 NEXT I 
7200 GOSUB 5010: IF HARD$="1" GOTO 7320: REM  If "Easy", don't move frog 
7210 IF FIN1=1 GOTO 7320: REM  Move fly after wrong answer 
7220 R=INT(RND*25): L=INT(RND*25) 
7230 IF FIN2=0 GOTO 7250 
7240 IF (R+PFLY)>65 OR (PFLY-L)<10 GOTO 7220 ELSE GOTO 7260 
7250 IF (R+PFLY)>65 OR (PFLY-L)<30 GOTO 7220 
7260 IF DIR=4 GOTO 7280 
7270 FOR J=1 TO R: PFLY=PFLY+1: GOSUB 6300: NEXT J 
7280 FOR J=1 TO L: PFLY=PFLY-1: GOSUB 6200: NEXT J 
7290 IF DIR=6 GOTO 7310 
7300 FOR J=1 TO R: PFLY=PFLY+1: GOSUB 6300: NEXT J 
7310 GOSUB 6100: REM  Stop Fly in new position 
7320 MSG=MSG+1: LOCATE 14,7,0: PRINT SPC(7) 
7330 LOCATE 15,1: PRINT SPC(37) 
7340 LOCATE 16,1: PRINT SPC(60) 
7350 LOCATE 17,1: PRINT SPC(60) 
7360 LOCATE 18,1: PRINT SPC(60) 
7370 RETURN 
7400 REM  Fire tongue of frog. 
7410 IF DIR=4 THEN GOSUB 5970 
7420 IF DIR=6 THEN GOSUB 5870 
7430 REM  Compare frog's location with that of spider and fly. 
7440 IF PSPD=(FROG+20) AND DIR=6 GOTO 7540 
7450 IF PSPD=(FROG-20) AND DIR=4 GOTO 7540 
7460 IF PSPD=(FROG+19) AND DIR=6 GOTO 7540 
7470 IF PSPD=(FROG-19) AND DIR=4 GOTO 7540 
7480 IF PFLY=(FROG+21) AND DIR=6 GOTO 7550 
7490 IF PFLY=(FROG-21) AND DIR=4 GOTO 7550 
7520 GOSUB 7200: REM Move fly after a miss. 
7530 COLOR 3,0: LOCATE 15,1: PRINT "Missed me!": GOTO 7200 
7540 COLOR 3,0: LOCATE 3,PSPD,0: PRINT "ZAPPED!": FIN2=1: RETURN 
7550 COLOR 3,0: LOCATE 3,PFLY,0: PRINT "ZAPPED!": FIN1=1: RETURN 

~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~~

If you have any questions or comments, please email me at:
z100lifeline@swvagts.com

Cheers,

Steven W. Vagts

